
Arbeitsblatt zur Integralrechnung
> restart;
> with(student):
> with(plots):
> f:=x->x^2;

Problem : Bestimme den Flächeninhalt der Fläache, die vom Graphen zu f(x)
und der x-Achse zwischen x = 2 und x = 4 eingeschlossen wird

Lösungsidee : Teile die Fläche in gleich breite Streifen ( im Bild 8 Stück ) so,
dass die linke Streifenecke jeweils Punkt des Graphen zu f(x) ist.

> pl2:=leftbox(f(x), x=2..4,8, color=RED,shading=yellow):
> pl3:=plot(f(x),x=-1..5,thickness=3):
> display(pl2,pl3);

Wenn die linke Intervallgrenze a, die rechte b heißt und n die Anzahl der Streifen,
dann ist die Streifenbreite dx :

> dx:=(a,b,n)->(b-a)/n;

Um den Flächeninhalt eines Streifens zu berechnen, muss man die Höhe des Streifens
und seine Breite dx haben. Die Höhe ist der Funktiosnwert an der richtigen Stelle x,
also für den 1. Streifen f(a), für den 2. Streifen  , für den 3. Streifen 

für die gesamte Fläche müssen alle Streifeninhalte aufsummiert werden.
Maple kann die Rechnung für beliebiges n ausführen :

> Au:=n->sum(f(a+i*dx(a,b,n))*dx(a,b,n),i=0..n-1);



> 'A'=Au(n);

Beim Berechnen wird eine Formel für Potenzsummen - siehe Formelsammlung - benötigt -
Maple kennt diese. Vereinfachen läßt sich der obige Term:

> 'A'=factor(Au(n));

Und jetzt der entscheidende Schritt: Die Anzahl der Streifen beliebig groß werden lassen :

> Limit(factor(Au(n)),n=infinity)=limit(Au(n),n=infinity);

Das Ergebnis ist bemerkenswert - der Flächeninhalt läßt sich berechnen,
wenn man für a und b konkrete Werte einsetzt, die Anzahl der Streifen spielt keine Rolle

Noch ist man nicht fertig - wenn man die Streifen wie im nächsten Bild konstruiert, sollte
der Flächeninhalt auch bestimmt werden können

> pl2:=rightbox(f(x), x=2..4,8, color=RED,shading=yellow):
> pl3:=plot(f(x),x=-1..5,thickness=3):
> display(pl2,pl3);



Das Verfahren wird modifiziert wiederholt, jetzt erhält man die Höhe der Streifen
als Funktionswert an der rechten Streifenkante, also  für den 1. Streifen usw.

> Ao:=n->sum(f(a+i*dx(a,b,n))*dx(a,b,n),i=1..n);

> 'A'=Ao(n);

> 'A'=factor(Ao(n));

> Limit(factor(Ao(n)),n=infinity)=limit(Ao(n),n=infinity);



Was man gehofft hat, tritt ein, die Werte für den Flächeninhalt stimmen überein,
der Flächeninhalt ist - durch passende Konstruktion - über eine Intervallschachtelung bestimmt

kleine Formelsammlung : Potenzsummen
1. Die Summe aller natürlicher Zahlen bis n , angefangen mit 1

> Sum(i,i=1..n)=factor(sum(i,i=1..n));

2. Die Summe aller Quadratzahlen bis n , angefangen mit 1

> Sum(i^2,i=1..n)=factor(sum(i^2,i=1..n));

3. Die Summe aller  bis n , angefangen mit 1

> Sum(i^3,i=1..n)=factor(sum(i^3,i=1..n));


